

COP820,COP840

AN-521 Dual Tone Multiple Frequency (DTMF)



Literature Number: SNOA782

Dual Tone Multiple Frequency (DTMF)

National Semiconductor
Application Note 521
Verne H. Wilson
June 1989



Dual Tone Multiple Frequency (DTMF)

The DTMF (Dual Tone Multiple Frequency) application is associated with digital telephony, and provides two selected output frequencies (one high band, one low band) for a duration of 100 ms. A benchmark subroutine has been written for the COP820C/840C microcontrollers, and is outlined in detail in this application note. This DTMF subroutine takes 110 bytes of COP820C/840C code, consisting of 78 bytes of program code and 32 bytes of ROM table. The timings in this DTMF subroutine are based on a 20 MHz COP820C/840C clock, giving an instruction cycle time of 1 μ s.

The matrix for selecting the high and low band frequencies associated with each key is shown in *Figure 1*. Each key is uniquely referenced by selecting one of the four low band frequencies associated with the matrix rows, coupled with selecting one of the four high band frequencies associated with the matrix columns. The low band frequencies are 697, 770, 852, and 941 Hz, while the high band frequencies are 1209, 1336, 1477, and 1633 Hz. The DTMF subroutine assumes that the key decoding is supplied as a low order hex digit in the accumulator. The COP820C/840C DTMF subroutine will then generate the selected high band and low band frequencies on port G output pins G3 and G2 respectively for a duration of 100 ms.

The COP820C/840C each contain only one timer. The problem is that three different times must be generated to satisfy the DTMF application. These three times are the periods of the two selected frequencies and the 100 ms duration period. Obviously the single timer can be used to generate any one (or possibly two) of the required times, with the program having to generate the other two (or one) times.

The solution to the DTMF problem lies in dividing the 100 ms time duration by the half periods (rounded to the nearest micro second) for each of the eight frequencies, and then examining the respective high band and low band quotients and remainders. The results of these divisions are detailed in Table I. The low band frequency quotients range from 139 to 188, while the high band quotients range from 241 to 326. The observation that only the low band quotients will each fit in a single byte dictates that the high band frequency be produced by the 16 bit (2 byte) COP820C/840C timer running in PWM (Pulse Width Modulation) Mode.




FIGURE 1. DTMF Keyboard Matrix

TL/DD/9662-1

The solution then is to use the program to produce the selected low band frequency as well as keep track of the 100 ms duration. This is achieved by using three programmed register counters R0, R2, and R3, with a backup register R1 to reload the counter R0. These three counters represent the half period, the 100 ms quotient, and the 100 ms remainder associated with each of the four low band frequencies.

The theory of operation in producing the selected low band frequency starts with loading the three counters with values obtained from a ROM table. The half period for the selected frequency is counted out, after which the G2 output bit is toggled. During this half period countout, the quotient counter is decremented. This procedure is repeated until the quotient counter counts out, after which the program branches to the remainder loop. During the remainder loop, the remainder counter counts out to terminate the 100 ms. Following the remainder countout, the G2 and G3 bits are both reset, after which the DTMF subroutine is exited. Great care must be taken in time balancing the half period loop for the selected low band frequency. Furthermore, the toggling of the G2 output bit (achieved with either a set or reset bit instruction) must also be exactly time balanced to maintain the half period time integrity. Local stall loops (consisting of a DRSZ instruction followed by a JP jump back to the DRSZ for a two byte, six instruction cycle loop) are embedded in both the half period and remainder loops. Consequently, the ROM table parameters for the half period and remainder counters are approximately only one sixth of what otherwise might be expected. The program for the half period loop, along with the detailed time balancing of the loop for each of the low band frequencies, is shown in *Figure 2*.

The DTMF subroutine makes use of two 16 byte ROM tables. The first ROM table contains the translation table for the input hex digit into the core vector. The encoding of the hex digit along with the hex digit ROM translation table is shown in Table II. The row and column bits (RR, CC) representing the low band and high band frequencies respectively of the keyboard matrix shown in *Figure 1*, are encoded in

TABLE I. Frequency Half Periods, Quotients, and Remainders

	Freq. Hz	Half Period 0.5P	Half Period in μ s	100 ms/0.5P	
				Quotient	Remainder
Low Band Freq.'s	697	717.36	717	139	337
	770	649.35	649	154	54
	852	586.85	587	170	210
	941	531.35	531	188	172
High Band Freq.'s	1209	413.56	414 (256 + 158)	241	226
	1336	374.25	374 (256 + 118)	267	142
	1477	338.52	339 (256 + 83)	294	334
	1633	306.18	306 (256 + 50)	326	244

AN-521

the two upper and two lower bits of the hex digit respectively. Consequently, the format for the hex digit bits is RRCC, so that the input byte in the accumulator will consist of 0000RRCC. The program changes this value into 1101RRCC before using it in setting up the address for the hex digit ROM translation table.

The core vectors from the hex digit ROM translation table consist of a format of XX00TT00, where the two T (Timer) bits select one of four high band frequencies, while the two X bits select one of four low band frequencies. The core vector is transformed into four different inputs for the second ROM table. This transformation of the core vector is shown in Table III. The core vector transformation produces a timer vector 1100TT00 (T), and three programmed coun-

ter vectors for R1, R2, and R3. The formats for the three counter vectors are 1100XX11 (F), 1100XX10 (Q), and 1100XX01 (R) for R1, R2, and R3 respectively. These four vectors produced from the core vector are then used as inputs to the second ROM table. One of these four vectors (the T vector) is a function of the T bits from the core vector, while the other three vectors (F, Q, R) are a function of the X bits. This correlates to only one parameter being needed for the timer (representing the selected high band frequency), while three parameters are needed for the three counters (half period, 100 ms quotient, 100 ms remainder) associated with the low band frequency and 100 ms duration. The frequency parameter ROM translation table, accessed by the T, F, Q, and R vectors, is shown in Table IV.

Program			Bytes/Cycle	Conditional Cycles		Cycles	Total Cycles
	LD	B, #PORTGD	2/3				
	LD	X, #R1	2/3				
LUP1:	LD	A, [X-]	1/3			3	
	IFBIT	2, [B]	1/1			1	
	JP	BYP1	1/3	3	1		
	X	A, [X+]	1/3		3		
	SBIT	2, [B]	1/1		1		
	JP	BYP2	1/3		3		
BYP1:	NOP		1/1	1			
	RBIT	2, [B]	1/1	1			
	X	A, [X+]	1/3	3			
BYP2:	DRSZ	R2	1/3 DECREMENT			3	
	JP	LUP2	1/3 Q COUNT			3	
	JP	FINI	1/3				
LUP2:	DRSZ	R0	1/3 DECREMENT		3	3	
	JP	LUP2	1/3 F COUNT		3	1	
	NOP		1/1			1	
	LD	A, [X]	1/3			3	
	IFEQ	A, #104	2/2			2	
	JP	LUP1	1/3		1	3	31
	NOP		1/1		1		
	IFEQ	A, #93	2/2		2		
BACK:	JP	LUP1	1/3	1	3		35
	JP	BACK	1/3	3			
				3			39

Table IV	×	Stall	+	Total	=	Half
Frequency		Loop		Cycles		Period
((114 - 1)		x 6)		+ 39		= 717
((104 - 1)		x 6)		+ 31		= 649
((93 - 1)		x 6)		+ 35		= 587
((83 - 1)		x 6)		+ 39		= 531

FIGURE 2. Time Balancing for Half Period Loop

TABLE II. Hex Digit ROM Translation Table

	0	1	2	3
ROW	697 Hz	770 Hz	852 Hz	941 Hz
COLUMN	1209 Hz	1336 Hz	1477 Hz	1633 Hz

ADDRESS	DATA (HEX)	KEYBOARD	
*			* HEX DIGIT IS RRCC,
0xD0	000	1	WHERE R = ROW #
0xD1	004	2	AND C = COLUMN #
0xD2	008	3	- - - EXAMPLE: KEY 3 IS ROW #0,
0xD3	00C	A	COLUMN #2, SO HEX DIGIT
0xD4	040	4	IS 0010 = 2
0xD5	044	5	RRCC
0xD6	048	6	
0xD7	04C	B	
0xD8	080	7	
0xD9	084	8	
0xDA	088	9	
0xDB	08C	C	
0xDC	0C0	*	
0xDD	0C4	0	
0xDE	0C8	#	
0xDF	0CC	D	

TABLE III. Core Vector Translation

CORE VECTOR	-	XX00TT00	-	-	-	-	-	-	-
									*
									* *
									* * *
TIMER VECTOR		TIMER	T						1100TT00
HALF PERIOD VECTOR		R1	F						1100XX11
QUOTIENT VECTOR		R2	Q						1100XX10
REMAINDER VECTOR		R3	R						1100XX01

TABLE IV. Frequency Parameter ROM Translation Table

T - TIMER F - FREQUENCY Q - QUOTIENT R - REMAINDER

ADDRESS	DATA (DEC)	VECTOR
0xC0	158	T
0xC1	53	R
0xC2	140	Q
0xC3	114	F
0xC4	118	T
0xC5	6	R
0xC6	155	Q
0xC7	104	F
0xC8	83	T
0xC9	32	R
0xCA	171	Q
0xCB	93	F
0xCC	50	T
0xCD	25	R
0xCE	189	Q
0xCF	83	F

In summary, the input hex digit selects one of 16 core vectors from the first ROM table. This core vector is then transformed into four other vectors (T, F, Q, R), which in turn are used to select four parameters from the second ROM table. These four parameters are used to load the timer, and the respective half period, quotient, and remainder counters. The first ROM table (representing the hex digit matrix table) is arbitrarily placed starting at ROM location 01D0, and has a reference setup with the ADD A, #0D0 instruction. The second ROM table (representing the frequency parameter table) must be placed starting at ROM location 01C0 (or 0xC0) in order to minimize program size, and has reference setups with the OR A, #0C3 instruction for the F vector and with the OR A, #0C0 instruction for the T vector.

The three parameters associated with the two X bits of the core vector require a multi-level table lookup capability with the LAID instruction. This is achieved with the following section of code in the DTMF subroutine:

```

LD      B, #R1
LD      X, #R4
X       A, [X]
LUP:   LD      A, [X]
        LAID
        X      A, [B+]
        DRSZ  R4
        IFBNE #4
        JP    LUP
    
```

This program code loads the F frequency vector into R4, and then decrements the vector each time around the loop. This successive loop decrementation of the R4 vector changes the F vector into the Q vector, and then changes the Q vector into the R vector. This R4 vector is used to access the ROM table with the LAID instruction. The X pointer references the R4 vector, while the B pointer is incremented each time around the loop after it has been used to store away the three selected ROM table parameters (one per loop). These three parameters are stored in sequential RAM locations R1, R2, and R3. The IFBNE test instruction is used to skip out of the loop once the three selected ROM table parameters have been accessed and stored away.

The timer is initialized to a count of 15 so that the first timer underflow and toggling of the G3 output bit (with timer PWM mode and G3 toggle output selected) will occur at the same time as the first toggling of the G2 output bit. The half period counts for the high band frequencies range from 306 to 414, so these values minus 256 are stored in the timer section of the second ROM table. The selected value from this frequency ROM table is then stored in the lower half of the timer autoreload register, while a 1 is stored in the upper half. The timer is selected for PWM output mode and started with the instruction LD [B], #0B0 where the B pointer is selecting the CNTRL register at memory location 0EE.

The DTMF subroutine for the COP820C/840C uses 110 bytes of code, consisting of 78 bytes of program code and 32 bytes of ROM table. A program routine to sequentially call the DTMF subroutine for each of the 16 hex digit inputs is supplied with the listing for the DTMF subroutine.

```
1          ;DTMF PROGRAM FOR COP820C/840C          VERNE H. WILSON
2          ;                                         5/1/89
3          ;DTMF - DUAL TONE MULTIPLE FREQUENCY
4          ;
5          ;PROGRAM NAME: DTMF.MAC
6          ;
7          .TITLE DTMF
8          .CHIP 840
9          ;***** THE DTMF SUBROUTINE CONTAINS 110 BYTES *****
10         ; ***** THE DTMF SUBROUTINE TIMES OUT IN 100MSEC *****
11         ; ** FROM THE FIRST TOGGLE OF THE G2/G3 OUTPUTS **
12         ; ** BASED ON A 20 MHZ COP820C/840C CLOCK **
13         ;
14         ;G PORT IS USED FOR THE TWO OUTPUTS
15         ; - HIGH BAND (HB) FREQUENCY OUTPUT ON G3
16         ; - LOW BAND (LB) FREQUENCY OUTPUT ON G2
17         ;
18         ;TIMER COUNTS OUT
19         ; - HB FREQUENCIES
20         ;
21         ;PROGRAM COUNTS OUT
22         ; - LB FREQUENCIES
23         ; - 100 MSEC DIVIDED BY LB HALF PERIOD QUOTIENT
24         ; - 100 MSEC DIVIDED BY LB HALF PERIOD REMAINDER
25         ;
26         ;FORMAT FOR THE 16 HEX DIGIT MATRIX VECTOR IS 1101RRCC,
27         ; WHERE - RR IS ROW SELECT (LB FREQUENCIES)
28         ; - CC IS COLUMN SELECT (HB FREQUENCIES)
29         ;
30         ;FORMAT FOR THE 16 CORE VECTORS FROM THE MATRIX SELECT
31         ; TABLE IS XX00TT00, WHERE - TT IS HB SELECT
32         ; - XX IS LB SELECT
33         ;
34         ;FREQUENCY VECTORS (HB & LB) FOR FREQ PARAMETER TABLE
35         ; MADE FROM CORE VECTORS
36         ;
37         ;HB FREQUENCY VECTORS(4) END WITH 00 FOR TIMER COUNTS,
38         ; WHERE VECTOR FORMAT IS 1100TT00
39         ;
40         ;LB FREQUENCY VECTORS(12) END WITH:
41         ; 11 FOR HALF PERIOD LOOP COUNTS,
42         ; WHERE VECTOR FORMAT IS 1100XX11
43         ; 10 FOR 100 MSEC DIVIDED BY HALF PERIOD QUOTIENTS,
44         ; WHERE VECTOR FORMAT IS 1100XX10
45         ; 01 FOR 100 MSEC DIVIDED BY HALF PERIOD REMAINDERS,
46         ; WHERE VECTOR FORMAT IS 1100XX01
47         ;
48         ;HEX DIGIT MATRIX TABLE AT HEX 01D* (OPTIONAL LOCATION,
49         ; DEPENDING ON 'ADD A,#0D0' INST. IMMEDIATE VALUE)
50         ;
51         ;FREQ PARAMETER TABLE AT HEX 01C* (REQUIRED LOCATION)
```

```

52          .FORM
53          ;
54          ;MAGIC:      CORE VECTOR
55          ;              XX00TT00
56          ;
57          ;   TIMER      T      TT00
58          ;   R1         F      XX11
59          ;   R2         Q      XX10
60          ;   R3         R      XX01
61          ;
62          ;DECLARATIONS:
63          00D0      PORTLD = 0D0      ; PORTL DATA REG
64          00D1      PORTLC = 0D1      ; PORTL CONFIG REG
65          00D4      PORTGD = 0D4      ; PORTG DATA REG
66          00D5      PORTGC = 0D5      ; PORTG CONFIG REG
67          00DC      PORTD = 0DC      ; PORTD REG
68          00EA      TIMERLO = 0EA     ; TIMER LOW COUNTER
69          00EE      CNTRL = 0EE      ; CONTROL REG
70          00EF      PSW = 0EF        ; PROC STATUS WORD
71          00F0      R0 = 0F0        ; LB FREQ LOOP COUNTER
72          00F1      R1 = 0F1        ; LB FREQ LOOP COUNT
73          00F2      R2 = 0F2        ; LB FREQ Q COUNT
74          00F3      R3 = 0F3        ; LB FREQ R COUNT
75          00F4      R4 = 0F4        ; LB FREQ TABLE VECTOR
76          ;
77          0000 DD2F  START:  LD      SP,#02F      ; HEX DIGIT MATRIX
78          0002 BCD1FF LD      PORTLC,#0FF      ; 1 2 3 A
79          0005 BCD080 LD      PORTLD,#080      ; 4 5 6 B
80          0008 DEDC  LD      B,#PORTD          ; 7 8 9 C
81          000A 9E00  LD      [B],#0          ; * 0 # D
82          000C AE    LOOP:  LD      A,[B]      ; DTMF TEST LOOP
83          000D 3160 JSR      DTMF          ; HEX MATRIX DIGIT
84          000F DEDC  LD      B,#PORTD          ; TO SUBROUTINE IS
85          0011 AE    LD      A,[B]      ; OUTPUT TO PORTD
86          0012 9405 ADD     A,#5          ; DO WILL TOGGLE
87          0014 A6    X        A,[B]      ; FOR EACH CALL OF
88          0015 6C    RBIT     4,[B]      ; DTMF SUBROUTINE
89          0016 9DD0 LD      A,PORTLD        ; PORTL OUTPUTS
90          0018 A1    SC        ; PROVIDE SYNC
91          0019 B0    RRC     A          ; OUTPUT ORDER IS
92          001A 9CDO X        A,PORTLD        ; 1,5,9,D,4,8,#,A,
93          001C EF    JP      LOOP          ; 7,0,3,B,*,2,6,C
94          ;
95          ;
96          ;

```

TL/DD/9662-3

```

97      0160      ;      .=0160
98      ;
99 0160 DED5      DTMF: LD      B, #PORTGC
100 0162 9B3F      LD      [B-], #03F
101 0164 6B        RBIT    3, [B]      ; OPTIONAL
102 0165 6A        RBIT    2, [B]      ; OPTIONAL
103      ;
104 0166 94D0      ADD     A, #0D0      ; DIGIT MATRIX TABLE
105 0168 A4        LAID
106      ;
107 0169 5F        LD      B, #0
108 016A A6        X      A, [B]
109 016B AE        LD      A, [B]
110 017B 65        SWAP   A
111 016C 97C3      OR     A, #0C3
112 016E DEF1      LD      B, #R1
113 0170 DCF4      LD      X, #R4
114 0172 B6        X      A, [X]
115 0173 BE        LUP:   LD      A, [X]
116 0174 A4        LAID
117 0175 A2        X      A, [B+]      ; LB FREQ TABLES
118 0176 C4        DRSZ   R4          ; (3 PARAMETERS)
119 0177 44        IFBNE  #4
120 0178 FA        JP     LUP
121      ;
122 0179 5F        LD      B, #0
123 017A AE        LD      A, [B]
124 017C 97C0      OR     A, #0C0
125 017E A4        LAID
126 017F DEEA      LD      B, #TIMERLO ; HB FREQ TABLE
127 0181 9A0F      LD      [B+], #15   ; (1 PARAMETER)
128 0183 9A00      LD      [B+], #0
129 0185 A2        X      A, [B+]
130 0186 9A01      LD      [B+], #1
131 0188 9EB0      LD      [B], #0B0   ; START TIMER PWM
132      ;
133 018A DED4      LD      B, #PORTGD
134 018C DCF1      LD      X, #R1
135      ;
136 018E BB        LUP1: LD      A, [X-]
137 018F 72        IFBIT  2, [B]      ; TEST LB OUTPUT
138 0190 03        JP     BYP1
139 0191 B2        X      A, [X+]
140 0192 7A        SBIT  2, [B]      ; SET LB OUTPUT
141 0193 03        JP     BYP2
142 0194 B8        BYP1: NOP
143 0195 6A        RBIT  2, [B]      ; RESET LB OUTPUT
144 0196 B2        X      A, [X+]
145 0197 C2        BYP2: DRSZ   R2          ; DECR. QUOT. COUNT
146 0198 01        JP     LUP2
147 0199 0C        JP     FINI      ; Q COUNT FINISHED
148      ;
149 019A C0        LUP2: DRSZ   R0          ; DECR. F COUNT
150 019B FE        JP     LUP2      ; LB (HALF PERIOD)
151      ;
152 019C B8        NOP
153 019D BE        LD      A, [X]      ; *****
154 019E 9268      IFEQ   A, #104      ; BALANCE
155 01A0 ED        JP     LUP1      ; LB FREQUENCY
156      ;      ; HALF PERIOD
157 01A1 B8        NOP      ; RESIDUE
158 01A2 925D      IFEQ   A, #93      ; DELAY FOR
159 01A4 E9        BACK: JP     LUP1      ; EACH OF 4
160 01A5 FE        JP     BACK     ; LB FREQ'S
161      ;      ; *****
162 01A6 C3        FINI: DRSZ   R3          ; DECR. REM. COUNT
163 01A7 FE        JP     FINI      ; R CNT NOT FINISHED
164      ;
165 01A8 BDEE6C    RBIT  4, CNTRL    ; STOP TIMER
166 01AB 6B        RBIT  3, [B]      ; CLR HB OUTPUT
167 01AC 6A        RBIT  2, [B]      ; CLR LB OUTPUT
168      ;
169 01AD 8E        RET
170      ;

```

```

171 .FORM
172 ;
173 ;FREQUENCY AND 100MSEC PARAMETER TABLE
174 01C0 . =01C0
175 ;
176 01C0 9E .BYTE 158 ; T
177 01C1 35 .BYTE 53 ; R
178 01C2 8C .BYTE 140 ; Q
179 01C3 72 .BYTE 114 ; F
180 01C4 76 .BYTE 118 ; T
181 01C5 06 .BYTE 6 ; R
182 01C6 9B .BYTE 155 ; Q
183 01C7 68 .BYTE 104 ; F
184 01C8 53 .BYTE 83 ; T
185 01C9 20 .BYTE 32 ; R
186 01CA AB .BYTE 171 ; Q
187 01CB 5D .BYTE 93 ; F
188 01CC 32 .BYTE 50 ; T
189 01CD 19 .BYTE 25 ; R
190 01CE BD .BYTE 189 ; Q
191 01CF 53 .BYTE 83 ; F
192 ;
193 ;DIGIT MATRIX TABLE
194 01D0 . =01D0
195 ;
196 01D0 00 .BYTE 000 ; 1 ROW COL
197 01D1 04 .BYTE 004 ; 2 0 0
198 01D2 08 .BYTE 008 ; 3 0 2
199 01D3 0C .BYTE 00C ; A 0 3
200 01D4 40 .BYTE 040 ; 4 1 0
201 01D5 44 .BYTE 044 ; 5 1 1
202 01D6 48 .BYTE 048 ; 6 1 2
203 01D7 4C .BYTE 04C ; B 1 3
204 01D8 80 .BYTE 080 ; 7 2 0
205 01D9 84 .BYTE 084 ; 8 2 1
206 01DA 88 .BYTE 088 ; 9 2 2
207 01DB 8C .BYTE 08C ; C 2 3
208 01DC C0 .BYTE 0C0 ; * 3 0
209 01DD C4 .BYTE 0C4 ; 0 3 1
210 01DE C8 .BYTE 0C8 ; # 3 2
211 01DF CC .BYTE 0CC ; D 3 3
212 ;
213 .END

```

SYMBOL TABLE

B	00FE	BACK	01A4	BYP1	0194	BYP2	0197
CNTRL	00EE	DTMF	0160	FINI	01A6	LOOP	000C
LUP	0174	LUP1	018E	LUP2	019A	PORTD	00DC
PORTGC	00D5	PORTGD	00D4	PORTLC	00D1	PORTLD	00D0
PSW	00EF *	R0	00F0	R1	00F1	R2	00F2
R3	00F3	R4	00F4	SP	00FD	START	0000 *
TIMERL	00EA	X	00FC				

MACRO TABLE

NO WARNING LINES

NO ERROR LINES

139 ROM BYTES USED

SOURCE CHECKSUM = 99A7
OBJECT CHECKSUM = 03E1

INPUT FILE C:DTMF.MAC
LISTING FILE C:DTMF.PRN
OBJECT FILE C:DTMF.LM

TL/DD/9662-6

The code listed in this App Note is available on Dial-A-Helper.

Dial-A-Helper is a service provided by the Microcontroller Applications Group. The Dial-A-Helper system provides access to an automated information storage and retrieval system that may be accessed over standard dial-up telephone lines 24 hours a day. The system capabilities include a MESSAGE SECTION (electronic mail) for communicating to and from the Microcontroller Applications Group and a FILE SECTION mode that can be used to search out and retrieve application data about NSC Microcontrollers. The minimum system requirement is a dumb terminal, 300 or 1200 baud modem, and a telephone.

With a communications package and a PC, the code detailed in this App Note can be down loaded from the FILE SECTION to disk for later use. The Dial-A-Helper telephone lines are:

Modem (408) 739-1162
Voice (408) 721-5582

For Additional Information, Please Contact Factory

LIFE SUPPORT POLICY

NATIONAL'S PRODUCTS ARE NOT AUTHORIZED FOR USE AS CRITICAL COMPONENTS IN LIFE SUPPORT DEVICES OR SYSTEMS WITHOUT THE EXPRESS WRITTEN APPROVAL OF THE PRESIDENT OF NATIONAL SEMICONDUCTOR CORPORATION. As used herein:

1. Life support devices or systems are devices or systems which, (a) are intended for surgical implant into the body, or (b) support or sustain life, and whose failure to perform, when properly used in accordance with instructions for use provided in the labeling, can be reasonably expected to result in a significant injury to the user.
2. A critical component is any component of a life support device or system whose failure to perform can be reasonably expected to cause the failure of the life support device or system, or to affect its safety or effectiveness.



National Semiconductor Corporation
 1111 West Bardin Road
 Arlington, TX 76017
 Tel: 1(800) 272-9959
 Fax: 1(800) 737-7018

National Semiconductor Europe
 Fax: (+49) 0-180-530 85 86
 Email: onjwge@tevm2.nsc.com
 Deutsch Tel: (+49) 0-180-530 85 85
 English Tel: (+49) 0-180-532 78 32
 Français Tel: (+49) 0-180-532 93 58
 Italiano Tel: (+49) 0-180-534 16 80

National Semiconductor Hong Kong Ltd.
 19th Floor, Straight Block,
 Ocean Centre, 5 Canton Rd.
 Tsimshatsui, Kowloon
 Hong Kong
 Tel: (852) 2737-1600
 Fax: (852) 2736-9960

National Semiconductor Japan Ltd.
 Tel: 81-043-299-2309
 Fax: 81-043-299-2408

IMPORTANT NOTICE

Texas Instruments Incorporated and its subsidiaries (TI) reserve the right to make corrections, modifications, enhancements, improvements, and other changes to its products and services at any time and to discontinue any product or service without notice. Customers should obtain the latest relevant information before placing orders and should verify that such information is current and complete. All products are sold subject to TI's terms and conditions of sale supplied at the time of order acknowledgment.

TI warrants performance of its hardware products to the specifications applicable at the time of sale in accordance with TI's standard warranty. Testing and other quality control techniques are used to the extent TI deems necessary to support this warranty. Except where mandated by government requirements, testing of all parameters of each product is not necessarily performed.

TI assumes no liability for applications assistance or customer product design. Customers are responsible for their products and applications using TI components. To minimize the risks associated with customer products and applications, customers should provide adequate design and operating safeguards.

TI does not warrant or represent that any license, either express or implied, is granted under any TI patent right, copyright, mask work right, or other TI intellectual property right relating to any combination, machine, or process in which TI products or services are used. Information published by TI regarding third-party products or services does not constitute a license from TI to use such products or services or a warranty or endorsement thereof. Use of such information may require a license from a third party under the patents or other intellectual property of the third party, or a license from TI under the patents or other intellectual property of TI.

Reproduction of TI information in TI data books or data sheets is permissible only if reproduction is without alteration and is accompanied by all associated warranties, conditions, limitations, and notices. Reproduction of this information with alteration is an unfair and deceptive business practice. TI is not responsible or liable for such altered documentation. Information of third parties may be subject to additional restrictions.

Resale of TI products or services with statements different from or beyond the parameters stated by TI for that product or service voids all express and any implied warranties for the associated TI product or service and is an unfair and deceptive business practice. TI is not responsible or liable for any such statements.

TI products are not authorized for use in safety-critical applications (such as life support) where a failure of the TI product would reasonably be expected to cause severe personal injury or death, unless officers of the parties have executed an agreement specifically governing such use. Buyers represent that they have all necessary expertise in the safety and regulatory ramifications of their applications, and acknowledge and agree that they are solely responsible for all legal, regulatory and safety-related requirements concerning their products and any use of TI products in such safety-critical applications, notwithstanding any applications-related information or support that may be provided by TI. Further, Buyers must fully indemnify TI and its representatives against any damages arising out of the use of TI products in such safety-critical applications.

TI products are neither designed nor intended for use in military/aerospace applications or environments unless the TI products are specifically designated by TI as military-grade or "enhanced plastic." Only products designated by TI as military-grade meet military specifications. Buyers acknowledge and agree that any such use of TI products which TI has not designated as military-grade is solely at the Buyer's risk, and that they are solely responsible for compliance with all legal and regulatory requirements in connection with such use.

TI products are neither designed nor intended for use in automotive applications or environments unless the specific TI products are designated by TI as compliant with ISO/TS 16949 requirements. Buyers acknowledge and agree that, if they use any non-designated products in automotive applications, TI will not be responsible for any failure to meet such requirements.

Following are URLs where you can obtain information on other Texas Instruments products and application solutions:

Products

Audio	www.ti.com/audio
Amplifiers	amplifier.ti.com
Data Converters	dataconverter.ti.com
DLP® Products	www.dlp.com
DSP	dsp.ti.com
Clocks and Timers	www.ti.com/clocks
Interface	interface.ti.com
Logic	logic.ti.com
Power Mgmt	power.ti.com
Microcontrollers	microcontroller.ti.com
RFID	www.ti-rfid.com
OMAP Mobile Processors	www.ti.com/omap
Wireless Connectivity	www.ti.com/wirelessconnectivity

Applications

Communications and Telecom	www.ti.com/communications
Computers and Peripherals	www.ti.com/computers
Consumer Electronics	www.ti.com/consumer-apps
Energy and Lighting	www.ti.com/energy
Industrial	www.ti.com/industrial
Medical	www.ti.com/medical
Security	www.ti.com/security
Space, Avionics and Defense	www.ti.com/space-avionics-defense
Transportation and Automotive	www.ti.com/automotive
Video and Imaging	www.ti.com/video

TI E2E Community Home Page

e2e.ti.com

Mailing Address: Texas Instruments, Post Office Box 655303, Dallas, Texas 75265
Copyright © 2011, Texas Instruments Incorporated